

Chapitre 2

Introduction au λ -calcul typé

On a vu que le λ -calcul est la pure théorie de la fonctionnalité. Mais on peut vouloir associer à chaque variable d'un terme un *type* qui indique sa nature fonctionnelle, c'est-à-dire ses domaine et codomaine. C'est naturel quand on pense aux mathématiques où on s'intéresse rarement à une fonction indépendamment de son ensemble de départ et de son ensemble d'arrivée. Les types permettent de retrouver cette idée dans le λ -calcul.

De plus, ils ont un intérêt calculatoire, car l'intérêt d'un système de types pour le λ -calcul (d'un point de vue théorique) est d'assurer la forte normalisation des termes typables. Ce sera le cas du système que nous allons voir, qui est le plus simple de tous. C'est pourquoi on l'appelle le système des types simples. Dans cette optique, le λ -calcul est vu comme un langage de programmation, et les types assurent la terminaison des programmes typables (indépendamment de la stratégie de réduction).

2.1 Les types simples pour le λ -calcul typé

La syntaxe du λ -calcul est identique à celle qui a été introduite précédemment.

Soit \mathcal{P} un ensemble de variables de types. On définit l'ensemble des types simples bien formés \mathcal{T} (construit à partir de \mathcal{P} et du constructeur \rightarrow) par :

- si $\alpha \in \mathcal{P}$ alors $\alpha \in \mathcal{T}$
- si $\sigma, \tau \in \mathcal{T}$ alors $\sigma \rightarrow \tau \in \mathcal{T}$.

Typier un terme (clos), c'est lui associer un type. Mais on ne peut pas se contenter de ne typer que les termes clos. Il faut donc se référer à un *contexte de types* (qui associe à chaque variable libre un type). Donc typer un terme (non nécessairement clos) consistera à lui associer un type dans un certain contexte où se trouvent toutes les variables libres du terme. Donnons une définition plus formelle.

Les contextes sont des listes de couples $C = (x_1 : \sigma_1), \dots, (x_n : \sigma_n)$.

On définit inductivement une relation $C \vdash M : \tau$ qui signifie " M est de type τ dans le contexte C " de la façon suivante :

- Si $C = (x_1 : \sigma_1), \dots, (x_n : \sigma_n)$, alors $C \vdash x_i : \sigma_i$ pour tout $i = 1, \dots, n$ (Var).
- Si $C \vdash M : \sigma \rightarrow \tau$ et $C \vdash N : \sigma$ alors $C \vdash MN : \tau$ (App).
- Et si $(x : \sigma)C \vdash M : \tau$ alors $C \vdash \lambda x.M : \sigma \rightarrow \tau$ (Abs).

Ces règles s'écrivent habituellement :

(Var)

$$(x_1 : \sigma_1), \dots, (x_n : \sigma_n) \vdash x_i : \sigma_i$$

(Abs)

$$\frac{(x : \sigma), C \vdash M : \tau}{C \vdash \lambda x.M : \sigma \rightarrow \tau}$$

Exemple de typage : on se propose de typer l'entier de Church "deux". C'est $\lambda f \lambda x.f(fx)$. On prouve qu'il admet le type $(\alpha \rightarrow \alpha) \rightarrow \alpha \rightarrow \alpha$ au moyens des règles ci-dessus.

- 1) on a $(x : \alpha), (f : \alpha \rightarrow \alpha) \vdash x : \alpha$
- 2) et $(x : \alpha), (f : \alpha \rightarrow \alpha) \vdash f : \alpha \rightarrow \alpha$
- 3) donc par 1) et 2) $(x : \alpha), (f : \alpha \rightarrow \alpha) \vdash fx : \alpha$
- 4) par 2) et 3) on a $(x : \alpha), (f : \alpha \rightarrow \alpha) \vdash f(fx) : \alpha$
- 5) puis par la règle d'abstraction, de 4) on déduit $(f : \alpha \rightarrow \alpha) \vdash \lambda x.f(fx) : \alpha \rightarrow \alpha$
- 6) et de même $(f : \alpha \rightarrow \alpha) \vdash \lambda f.\lambda x.f(fx) : (\alpha \rightarrow \alpha) \rightarrow \alpha \rightarrow \alpha$

Ce qui s'écrit plus facilement en utilisant la deuxième forme des règles :

$$\frac{\frac{x : \alpha, f : \alpha \rightarrow \alpha \vdash f : \alpha \rightarrow \alpha \quad x : \alpha, f : \alpha \rightarrow \alpha \vdash x : \alpha}{x : \alpha, f : \alpha \rightarrow \alpha \vdash fx : \alpha}}{x : \alpha, f : \alpha \rightarrow \alpha \vdash f(fx) : \alpha}}{f : \alpha \rightarrow \alpha \vdash \lambda x.f(fx) : \alpha \rightarrow \alpha}}{\vdash \lambda f.\lambda x.f(fx) : (\alpha \rightarrow \alpha) \rightarrow \alpha \rightarrow \alpha}$$

On a donné une méthode de *vérification* de typage, mais pas de *synthèse* d'un type.

Pour ce système de types, il existe une méthode de synthèse utilisant l'unification, que nous allons donner. En voici un avant-goût par le raisonnement intuitif suivant :

On suppose que dans ce terme, f a le type σ et x a le type τ . Comme on a pu former fx il faut que σ soit de la forme $\tau \rightarrow \sigma_1$ et alors fx est du type σ_1 . Comme on a pu former $f(fx)$, il faut que $\tau = \sigma_1$. En conclusion, $f : \tau \rightarrow \tau$ et $x : \tau$, et le terme a le type $(\tau \rightarrow \tau) \rightarrow \tau \rightarrow \tau$ pour n'importe quel type τ (en particulier, on peut prendre une variable α).

2.2 Synthèse par unification

2.2.1 Unification

On va rappeler l'algorithme d'unification, dans le cas simple qui est le nôtre. On considère une signature ne comportant qu'un seul symbole, une fonction a d'arité 2 (c'est la flèche \rightarrow des types). Les variables que nous considérerons sont celles de l'ensemble \mathcal{P} .

Rappelons qu'une substitution est une application $\mathcal{P} \rightarrow \mathcal{T}$ (\mathcal{T} est l'ensemble des termes bâtis sur cette signature), et qu'on peut appliquer une substitution s à un terme $\sigma \in \mathcal{T}$, ce qui donne un terme σs :

- $x s = s(x)$
- $(a(\sigma, \tau))s = a(\sigma s, \tau s)$

Cela permet de définir la composée $s \circ t$ de deux substitutions comme une substitution.

Le problème consiste à résoudre des équations dans cette signature, et de trouver la solution la plus générale possible. Pour donner un sens à cela, il faut mettre un ordre sur les substitutions.

Définition 1 *On dit que s est plus général que t s'il existe une substitution u telle que $u \circ s = t$.*

Si $\sigma, \tau \in \mathcal{T}$, unifier σ et τ , c'est trouver une substitution s telle que $\sigma s = \tau s$. Alors le résultat est le suivant :

Théorème 1 *Pour tous $\sigma, \tau \in \mathcal{T}$ unifiable, il existe un unificateur qui est plus général que tous les autres. On l'appelle unificateur principal, et on le note $UP(\sigma, \tau)$.*

La preuve de ce théorème ne présente d'intérêt que par l'algorithme d'unification qu'elle propose, qui calcule cet unificateur principal ou échoue. Nous rappelons à présent cet algorithme. On remarquera qu'ici la seule cause d'échec possible est celle qui résulte du test d'occurrence d'une variable.

On notera $[\sigma/\alpha]$ la substitution qui envoie α sur σ et toutes les autres variables sur elles-mêmes. Alors, on définit la fonction UP qui prend deux termes et rend une substitution inductivement par :

- $UP(\alpha, \sigma) = [\sigma/\alpha]$ si α ne figure pas dans σ , et echec si non.
- $UP(\sigma, \alpha) = [\sigma/\alpha]$ si α ne figure pas dans σ , et echec si non.
- si $s = UP(\sigma, \sigma')$ et $t = UP(\tau s, \tau' s)$, alors $UP(a(\sigma, \tau), a(\sigma', \tau')) = t \circ s$.

Justification de l'algorithme (ce n'est pas une preuve rigoureuse). Le seul cas non trivial est celui où l'on a à unifier $a(\sigma_1, \tau_1)$ avec $a(\sigma_2, \tau_2)$. Alors si $s = UN(\sigma_1, \sigma_2)$, alors $\sigma_1 s = \sigma_2 s$ et si $t = UN(\tau_1 s, \tau_2 s)$, alors $(\tau_1 s)t = (\tau_2 s)t$, c'est-à-dire $\tau_1 u = \tau_2 u$ où $u = t \circ s$. On voit facilement que u est un unificateur de nos deux termes de départ.

Voici l'algorithme d'unification dans le cas des types simples. La syntaxe abstraite des types :

```
#type Type = Tvar of string | Arrow of Type * Type ;;
Type Type defined.
```

On représente les substitutions par des listes de couples chaîne de caractère, terme, Voici l'application d'une substitution à un type :

```
#let substitute s =
  let rec sub_rec = function
    (Tvar a) as alpha  $\rightarrow$  (try assoc a s with Not_found  $\rightarrow$  alpha)
  | Arrow(T1,T2)  $\rightarrow$  Arrow(sub_rec T1, sub_rec T2)
  in sub_rec
  ;;
substitute : (string * Type) list  $\rightarrow$  Type  $\rightarrow$  Type = <fun>
```

et voici la composition de deux substitutions :

```
#let compsubst s t =
  (map (function (a,T)  $\rightarrow$  (a,substitute s T)) t) @ s ;;
compsubst : (string * Type) list  $\rightarrow$  (string * Type) list  $\rightarrow$  (string * Type) list
           = <fun>
```

La fonction `occur` teste l'existence d'une variable dans un type.

```
#let occur a =
  let rec occ_rec = function
    Tvar b  $\rightarrow$  b=a
  | Arrow(T1,T2)  $\rightarrow$  occ_rec T1 or occ_rec T2
  in occ_rec
  ;;
occur : string  $\rightarrow$  Type  $\rightarrow$  bool = <fun>
```

La fonction `unify` calcule l'unificateur principal pour deux types simples. Elle retourne cet unificateur quand il existe ou échoue en déclenchant l'exception `Failure "unify"`.

```
#let rec unify = function
  (Tvar a), T  $\rightarrow$ 
    if T=Tvar a then []
    else if occur a T then raise (Failure "unify")
    else [a,T]
  | T,Tvar a  $\rightarrow$  if occur a T then raise (Failure "unify")
    else [a,T]
  | Arrow(S1,S2), Arrow(T1,T2)  $\rightarrow$ 
    let s=unify(S1,T1)
```

```

in
  compsubst (unify(substitute s S2,
                  substitute s T2)) s;;
unify : Type * Type → (string * Type) list = <fun>

```

Les exemples suivants calculent l'unificateur principal de deux types. Cette substitution est ensuite appliquée aux types pour vérifier leur égalité après substitution.

1. Un des cas les plus simples où une variable de type est substituée par une autre :

```

#let T1=(Tvar "x");;
T1 : Type = Tvar "x"

#let T2=(Tvar "y");;
T2 : Type = Tvar "y"

#let s1 = unify (T1,T2);;
s1 : (string * Type) list = ["x", Tvar "y"]

#substitute s1 T1 = substitute s1 T2;;
- : bool = true

```

2. Un cas classique où une variable est substituée par un type fonctionnel :

```

#let T3 = (Arrow (Tvar "a", Arrow (Tvar "b",Tvar "c")));;
T3 : Type = Arrow (Tvar "a", Arrow (Tvar "b", Tvar "c"))

#let T4 = (Arrow (Arrow (Tvar "d",Tvar "e"), (Arrow (Tvar "f",Tvar "g"))));;
T4 : Type = Arrow (Arrow (Tvar "d", Tvar "e"), Arrow (Tvar "f", Tvar "g"))

#let s2 = unify (T3,T4);;
s2 : (string * Type) list = ["a", Arrow (Tvar "d", Tvar "e"); "b", Tvar "f";
                             "c", Tvar "g"]

#substitute s2 T3;;
- : Type = Arrow (Arrow (Tvar "d", Tvar "e"), Arrow (Tvar "f", Tvar "g"))

#substitute s2 T4;;
- : Type = Arrow (Arrow (Tvar "d", Tvar "e"), Arrow (Tvar "f", Tvar "g"))

```

3. Un exemple où l'unification fait disparaître presque toutes les variables :

```

#let T5 = (Arrow (Arrow (Tvar "d",Tvar "a"), (Arrow (Tvar "a",Tvar "g"))));;
T5 : Type = Arrow (Arrow (Tvar "d", Tvar "a"), Arrow (Tvar "a", Tvar "g"))

#let T6 = (Arrow (Arrow (Tvar "a",Tvar "e"), (Arrow (Tvar "a",Tvar "d"))));;
T6 : Type = Arrow (Arrow (Tvar "a", Tvar "e"), Arrow (Tvar "a", Tvar "d"))

#let s3 = unify (T5,T6);;
s3 : (string * Type) list = ["d", Tvar "e"; "a", Tvar "e"; "g", Tvar "e"]

```

```
#substitute s3 T5 = substitute s3 T6 ;;
- : bool = true
```

4. Enfin un cas d'échec :

```
#unify (T3, T5);;
Uncaught exception: Failure "unify"
```

On peut maintenant donner l'algorithme de typage.

2.2.2 Typage

On définit une fonction T qui prend un contexte et un terme M et rend un couple constitué d'une substitution et du type cherché pour M dans ce contexte (on produit le type le plus général possible).

Cette substitution rendue par l'algorithme s'applique aux variables de type du contexte, et contient les contraintes produites par le typage du terme.

- $T(x, C) = (C(x), id)$
- Pour calculer $T(\lambda x.M, C)$, on introduit une nouvelle variable de type α , et on essaie de typer M dans le nouveau contexte $(x : \alpha)C$; soit $(\sigma, s) = T(M, (x : \alpha)C)$. Alors le type cherché est $s(\alpha) \rightarrow \sigma$, et on rend la substitution s .
- Pour calculer $T(MN, C)$, on calcule $(\sigma, s) = T(M, C)$, et $(\tau, t) = T(N, Cs)$, car il faut tenir compte des contraintes induites par le typage de M . On veut pouvoir appliquer M à N , et pour cela il faut que M ait un type fonctionnel dans le contexte courant, c'est-à-dire que σt soit de la forme $\tau \rightarrow \phi$ où ϕ est un type à déterminer. Pour cela, on calcule l'unificateur principal u de σt et $\tau \rightarrow \alpha$ où α est une nouvelle variable de type. On rend le type $u(\alpha)$, et la substitution $u \circ t \circ s$.

Voici le programme de typage en Caml :

Le type `term` pour la représentation des λ -termes et la fonction `gensym` pour la création des variables de type sont définis ainsi :

```
#type term = Var of string
           | Abs of string * term
           | App of term * term ;;
```

Type term defined.

```
#let gensym =
  let c = ref 0 in
  function s → (c := !c+1 ; s^(string_of_int !c)) ;;
gensym : string → string = <fun>
```

La fonction suivante `TYPE` suit très exactement l'algorithme de typage indiqué ci-dessus.

```

#let rec TYPE C = function
  Var x → assoc x C , [ ]
| Abs(x,M) → let alpha=gensym "a"
              in
              let (sigma,s) = TYPE ((x,Tvar alpha)::C) M
              in
              Arrow( (try assoc alpha s with _ → Tvar alpha),
                    sigma), s
| App(M,N) → let (sigma,s)=TYPE C M
              in
              let (tau,t) = TYPE (map (function (x,phi) →
                                          (x,substitute s phi)) C) N
              in
              let alpha=gensym "a"
              in
              let u= unify(substitute t sigma, Arrow(tau,Tvar alpha))
              in
              (try assoc alpha u with _ → Tvar alpha),
              comsubst u (comsubst t s)
;;
TYPE : (string * Type) list → term → Type * (string * Type) list = <fun>

```

La fonction `print_type` donne un affichage lisible des types.

```

#let print_type t =
  let rec ptype = function
    Tvar x → print_string x
  | Arrow (x,y) → print_string "("; ptype x; print_string " -> ";
                ptype y; print_string ")"
  in
  ptype t; print_newline()
;;
print_type : Type → unit = <fun>

```

Les exemples suivants reprennent des termes déjà définis.

1. *A* :

```

#let A = Abs ( "x" , Abs ("y" , App (Var "x", Var "y") ) );
A : term = Abs ("x", Abs ("y", App (Var "x", Var "y")))

#TYPE [ ] A ;;
- : Type * (string * Type) list = Arrow (Arrow (Tvar "a2", Tvar "a3"),
                                          Arrow (Tvar "a2", Tvar "a3")),
["a1", Arrow (Tvar "a2", Tvar "a3")]

```

```

#print_type (fst (TYPE [ ] A));;
((a5 -> a6) -> (a5 -> a6))
- : unit = ()

2. S :
#let S = Abs( "x" ,
             Abs ( "y" ,
                  Abs ( "z", App ( App (Var "x" , Var "z" ) ,
                                   App (Var "y", Var "z") ) ) ) );;
S : term = Abs ("x", Abs ("y", Abs ("z", App (App (Var "x", Var "z"),
                                                App (Var "y", Var "z")))))

#print_type (fst (TYPE [ ] S));;
((a9 -> (a11 -> a12)) -> ((a9 -> a11) -> (a9 -> a12)))
- : unit = ()

3.  $\Delta$  :
#let delta = Abs ( "x" , App (Var "x" , Var "x" ) );;
delta : term = Abs ("x", App (Var "x", Var "x"))

#print_type (fst (TYPE [ ] delta));;
Uncaught exception: Failure "unify"

```

2.3 Normalisation

On dit qu'un terme M est typable (avec les types simples) s'il existe un contexte C et un type σ tels que $C \vdash M : \sigma$. Voici le théorème :

Théorème 2 *Tout terme typable (avec les types simples) est fortement normalisable.*

On n'en donnera pas la preuve.

2.4 Polymorphisme

Le système des types simples ne donne pas le vrai polymorphisme malgré son ensemble de variables de type. En effet le terme $(\lambda f.f f)(\lambda x.x)$ n'est pas typable car la variable f prend deux valeurs : $\alpha \rightarrow \alpha$ et $(\alpha \rightarrow \alpha) \rightarrow \alpha \rightarrow \alpha$. Ce qui est impossible car une variable ne possède qu'un seul type.

Pour cela on étend le système de type. Soit \mathcal{P} un ensemble de variables de types, \mathcal{T} l'ensemble des types simples construit à partir de \mathcal{T} et \mathcal{S} l'ensemble des schémas de types construit à partir de \mathcal{P} et \mathcal{T} .

- types simples :
- si $\alpha \in \mathcal{P}$ alors $\alpha \in \mathcal{T}$
- si $\sigma, \tau \in \mathcal{T}$ alors $\sigma \rightarrow \tau \in \mathcal{T}$.

- schémas de type :
- si $\tau \in \mathcal{T}$ alors $\tau \in \mathcal{S}$
- si $\sigma \in \mathcal{S}, \alpha \in \mathcal{P}$ alors $\forall \alpha. \sigma \in \mathcal{S}$

On obtient ainsi de nouvelles règles de typage :

(Var)

$$(x_1 : \sigma_1), \dots, (x_n : \sigma_n) \vdash x_i : \tau[\tau_i/\alpha_i] \quad \sigma_i = \forall \alpha_1, \dots, \alpha_n. \tau$$

(App)

$$\frac{C \vdash M : \tau \rightarrow \tau' \quad C \vdash N : \tau}{C \vdash MN : \tau'}$$

(Abs)

$$\frac{(x : \tau), C \vdash M : \tau'}{C \vdash \lambda x. M : \tau \rightarrow \tau'}$$

(Let)

$$\frac{C \vdash N : \tau \quad \alpha_1, \dots, \alpha_n = V(\tau) - V(C) \quad (x : \forall \alpha_1, \dots, \alpha_n. \tau) \quad C \vdash M : \tau'}{C \vdash \text{let } x = N \text{ in } M : \tau'}$$

Ce qui nous permet de typer `let f = $\lambda x. x$ in ff` :

$$\frac{\frac{\frac{VAR}{f : \forall \beta. \beta \rightarrow \beta \vdash f : (\alpha \rightarrow \alpha) \rightarrow \alpha \rightarrow \alpha}}{\vdash \lambda x. x : \beta \rightarrow \beta} \quad \beta = V(\beta \rightarrow \beta) - V(\emptyset) \quad \frac{VAR}{f : \forall \beta. \beta \rightarrow \beta \vdash f : \alpha \rightarrow \alpha}}{f : \forall \beta. \beta \rightarrow \beta \vdash ff : \alpha \rightarrow \alpha}}{\vdash \text{let } f = \lambda x. x \text{ in } ff : \alpha \rightarrow \alpha}$$

2.5 Exercices

1. Vérifier le type de `S`.
2. Écrire une fonction `new_print_type` qui n'affiche pas la représentation interne du type, comme celle indiquée ci-dessus, mais une représentation externe à la manière de Caml. Par exemple :

```
new_print_type (fst (TYPE [] A));;
```

```
affichera ('a -> 'b) -> ('a -> 'b).
```

3. Typier : `let deux = $\lambda f. \lambda x. f(fx)$ in deux deux`

2.6 Bibliographie

- R. Milner. "A theory of type polymorphism in programming" J. Comput. Syst. Sci. 1978

Premier article où le typage polymorphe de ML est décrit.

- Jean-Louis Krivine. λ -calcul, types et modèles, 1990. Masson.

Un classique par un spécialiste.

et les ouvrages cités dans la bibliographie sur le λ -calcul parlant du typage.